
Connecting 2 Arduinos by Bluetooth using 
a HC-05 and a HC-06: Pair, Bind, and Link 
Posted on November 25, 2014  

In a previous post, I showed how the HC-05 can auto connect to other Bluetooth devices by setting 

the HC-05 to pair with any device using CMODE=1. This is quick and easy but does not give any 

control over which another device the HC-05 connects to. 

In this post, I show how to set up the HC-05 to always connect to the same HC-06. For this we use 

PAIR, BIND, and LINK. 

I am using 2 separate Arduino IDEs; version 1.6.3 which is installed, and version 1.0.5 which I run 

from a folder (it is the non-install version). This allows me to use 2 IDEs at the same time, each 

connected to a different Arduino. It also gives me 2 serial monitors, one for each Arduino. 

You do not need to use different versions of the IDE. You can run 2 instances of the same IDE. 

Thanks to Brian for the tip. 

The modules are the zs-040 versions of the HC-05 and the HC-06.  

The HC-05 has 2 AT modes which I refer to as “mini” AT mode and “full” AT mode and some 

commands only work when in “full” AT mode. To enter “full” AT mode pin 34 needs to be HIGH and 

kept HIGH. To accomplish this I have made a connection from pin 34 to +3.3v. See the diagram 

below (or after the jump). 

If you are not sure about the different AT modes take a look at Arduino with HC-05 (ZS-040) 

Bluetooth module – AT MODE 

If you are not familiar with how the HC-06 and HC-05 work it may be worth while checking out some 

of the other posts: 

HC-05 and HC-06 zs-040 Bluetooth modules 

Arduino and HC-06 (ZS-040) 

Arduino With HC-05 Bluetooth Module in Slave Mode 

Connecting 2 Arduinos by Bluetooth using a HC-05 and a HC-06: Easy Method Using CMODE 

Connections 

I have 2 Arduinos. One connected to a HC-05 zs-040 and one connected to a HC-06 zs-040.  

The HC-05 has a connection from pin 34 to +3.3v. This activates “full” AT mode. 

HC-05 Set Up 

I have the HC-05 set with the following values: 

– ROLE = 0 (slave mode) 

– UART = 9600 (baud rate for communication mode) 

– CMODE = 0 (only connect to paired devices) 

– PSWD = 1234 (password/PIN for pairing) 

In AT mode I will be using a baud rate of 38400. 

Build the circuit, power on and load the following sketch. 

// Basic Bluetooth sketch HC-05_02_38400 

// Connect the HC-05 module and communicate using the serial monitor 

// 

// The HC-05 defaults to commincation mode when first powered on. 

// Needs to be placed in to AT mode 

// After a factory reset the default baud rate for communication mode is 38400 

// 

// 

//  Pins 

//  BT VCC to Arduino 5V out.  

//  BT GND to GND 

//  BT RX to Arduino pin 3 (through a voltage divider) 

//  BT TX to Arduino pin 2 (no need voltage divider) 

  

  

#include <SoftwareSerial.h> 

SoftwareSerial BTserial(2, 3); // RX | TX 

// Connect the HC-05 TX to Arduino pin 2 RX.  

// Connect the HC-05 RX to Arduino pin 3 TX through a voltage divider. 

  

char c = ' '; 

  

void setup()  

{ 

    // start th serial communication with the host computer 

    Serial.begin(9600); 

    Serial.println("Arduino with HC-05 is ready"); 

  

    // start communication with the HC-05 using 38400 

    BTserial.begin(38400);   

    Serial.println("BTserial started at 38400"); 

} 

  

void loop() 

{ 

  

     // Keep reading from HC-05 and send to Arduino Serial Monitor 

    if (BTserial.available()) 

    {   

        c = BTserial.read(); 

        Serial.write(c); 

    } 

  

    // Keep reading from Arduino Serial Monitor and send to HC-05 

    if (Serial.available()) 

    { 

        c =  Serial.read(); 

  

        // mirror the commands back to the serial monitor 

        // makes it easy to follow the commands 

        Serial.write(c);    

        BTserial.write(c);   

    } 

  

} 

Because pin 34 on the Bluetooth module is brought HIGH, the HC-05 should enter AT mode on 

power on. 

Open the serial monitor on the host computer and confirm you are in AT mode by typing “AT” (no 

quotes). You should get a reply of “OK” 

Remember that the HC-05 requires “Both NL & CR” to be set. 

The sketch echoes the commands you enter back to the serial monitor. This makes it a bit easier to 

follow. 

 

HC-06 Set Up 

Build the HC-06 circuit, power on and load the following sketch. My HC-06 is set to use 9600 baud 

rate. If yours is using a different speed then you will need to change the sketch. Change the 

following lines: 
BTSerial.begin(9600); 

Serial.println(“BTserial started at 9600″); 

Change the 9600 to what ever speed your HC-06 is using. 

// Basic bluetooth test sketch. HC-06_01_9600 

// HC-06 ZS-040  

//  

//  

//  Uses hardware serial to talk to the host computer and software serial for communication with the bluetooth module 

// 

//  Pins 

//  BT VCC to Arduino 5V out.  

//  BT GND to GND 

//  BT RX to Arduino pin 3 (through a voltage divider) 

//  BT TX to Arduino pin 2 (no need voltage divider) 

// 

//  When a command is entered in the serial monitor on the computer  

//  the Arduino will relay it to the bluetooth module and display the result. 

http://www.martyncurrey.com/wp-content/uploads/2015/07/HC-05_to_HC-06-HC-05_AT_2.jpg


http://www.martyncurrey.com/connecting-2-arduinos-by-bluetooth-using-a-hc-05-and-a-hc-06-
pair-bind-and-link/ 

Page 5 of 24 

// 

//  These HC-06 modules require capital letters and no line ending 

// 

  

#include <SoftwareSerial.h> 

SoftwareSerial BTSerial(2, 3); // RX | TX 

  

void setup()  

{ 

    Serial.begin(9600); 

    Serial.println("Arduino with HC-06 is ready"); 

  

    // HC-06 default baud rate is 9600 

    BTSerial.begin(9600);   

    Serial.println("BTserial started at 9600"); 

} 

  

void loop() 

{ 

  

  // Keep reading from HC-06 and send to Arduino Serial Monitor 

  if (BTSerial.available()) 

    Serial.write(BTSerial.read()); 

  

  // Keep reading from Arduino Serial Monitor and send to HC-06 

  if (Serial.available()) 

  BTSerial.write(Serial.read()); 

} 

HC-06 modules start in AT mode so we can start entering commands as soon as we open the serial 

monitor. Remember that the HC-06 requires commands to be upper case and does not want new 

lines or carriage returns so set “No line endings” at the bottom of the serial monitor window. Open 

the serial monitor on the host computer and confirm you are in AT mode by typing “AT” (no quotes). 

You should get a reply of “OK” 

 

We don’t need to do much to prepare the HC-06. Just make sure the password is the same as the 

HC-05. 

To change the password/PIN to 1234 enter “AT+PSWD1234″ (no quotes) and click Send 

http://www.martyncurrey.com/wp-content/uploads/2015/07/HC-05_to_HC-06-HC-06_AT.jpg

Pairing and Binding 

At this point we should have the HC-06 waiting for a connection. The password/PIN is 1234.  

The HC-05 should be in “full” AT mode. If you are not sure what this is see Arduino with HC-05 (ZS-

040) Bluetooth module – AT MODE 

The steps to connect to the HC-06 are 

1. set the same baud rate on both devices. 

2. make sure the passwords on the HC-05 and the HC-06 are the same 

3. find the address of the HC-06 

4. pair the HC-05 with the HC-06 

5. bind the HC-06 to the HC-05 

6. set the HC-05 to only connect with paired devices 

7. link to the HC-06 

1. Set the same baud rate on both devices. 

Set the communication mode baud rate to 9600 on the HC-05 and the HC-06. I have already done 

this. 

This is only for the purpose of this example. Once you have got everything working you can change 

the baud rates. 

2. Make sure the passwords on the HC-05 and the HC-06 are the same 

We have already set the passwords. 

3. Find the address of the HC-06 

You can do this with an Android device or you can do it with the HC-05. Here is how to it with the 

HC-05. 

Make sure the HC-05 is in AT mode and enter the following commands. 

AT+RMAAD 

AT+ROLE=1 

AT+RESET 

http://www.martyncurrey.com/arduino-with-hc-05-bluetooth-module-at-mode/
http://www.martyncurrey.com/arduino-with-hc-05-bluetooth-module-at-mode/
http://www.martyncurrey.com/wp-content/uploads/2015/07/HC-05_to_HC-06-HC-06_setPin.jpg

AT+RMAAD clears any previously paired devices. 

AT+ROLE=1 puts the HC-05 in Master Mode 

AT+RESET reset the HC-05. This is sometimes needed after changing roles.  

Now enter the following 

AT+CMODE=0 

AT+INQM=0,5,9 

AT+INIT 

AT+INQ 

AT+INQM=0,5,9, AT+INIT and AT+INQ require pin 34 to be HIGH. If pin 34 is not HIGH then you will 

not receive any reply from these commands. No message, no error message. 

 

AT+CMODE=0 allows the HC-05 to connect to any device 

AT+INQM=0,5,9 set inquiry to search for up to 5 devices for 9 seconds 

AT+INIT initiates the SPP profile. If SPP is already active you will get an error(17) which you can 

http://www.martyncurrey.com/wp-content/uploads/2015/07/HC-05_to_HC-06-HC-05_Connect01.jpg
http://www.martyncurrey.com/wp-content/uploads/2015/07/HC-05_to_HC-06-HC-05_Connect02.jpg

ignore. 

AT+INQ searches for other Bluetooth devices. 

In the above screen shot you can see that the HC-05 found 2 Bluetooth devices. 

+INQ:A854:B2:3FB035,8043C,7FFF 

+INQ:3014:10:171179,1F00,7FFF 

One of these is the HC-06. To find out we can use the AT+RNAME? command.  

AT+INQ returns 3 values, the first is the address and this is the value we need. The second value is 

the class of the device and the third value is the signal strength (RSSI). 

AT+INQ will only work if the HC-05 is in Master Mode and after a AT+INIT command. 

To get the name of the found devices we use the AT+RNAME? command. The address is the first 

field returned by AT+INQ; A854:B2:3FB035 and 3014:10:171179. 

When we enter the address we need to replace the colons with commas. The format is 

AT+RNAME?A854,B2,3FB035 and AT+RNAME?3014,10,171179 

Of course, you would use the address of your HC-06. 

 

You can see from the above. A854:B2:3FB035 is my TV and 3014:10:171179 is the HC-06. 

Now we have the address of the HC-06 we can finish making the connection. 

4. Pair the HC-05 with the HC-06 

To pair we use the AT+PAIR=<addr>,<timeout> command. 

Enter “AT+PAIR=3014,10,171179,9″ (no quotes) 

If the HC-05 cannot pair with the HC-06 within 9 seconds you will get an error message. If the pairing 

is successful, you will get an “OK” 

5. Bind the HC-06 to the HC-05 

Bind using “AT+BIND=3014,10,171179″ 

http://www.martyncurrey.com/wp-content/uploads/2015/07/HC-05_to_HC-06-HC-05_Connect03.jpg

6. Set the HC-05 to only connect with paired devices 

We do this with the CMODE command – “AT+CMODE=1″ 

7. Link to the HC-06 

Use the link command AT+LINK=<addr> 

In my case “AT+LINK=3014,10,171179″. If all is well you will get the “OK” reply. 

 

The LED on the HC-05 should have 2 quick blinks every 2 seconds (or so). 

The LED on the HC-06 should be on (not blinking). 

You have now connected the HC-05 and the HC-06 

Now that the connection has been formed, the HC-05 will automatically connect to the HC-06 every 

time they are turned on. 

Once the connection is formed the HC-05 will switch to communication mode. If you want to 

continue with AT Mode, then you will need to reset the module with pin 34 HIGH or with the button 

switch closed. 

http://www.martyncurrey.com/wp-content/uploads/2015/07/HC-05_to_HC-06-HC-05_Connect04.jpg

Test the Connection 

On the HC-05, remove the connection to pin 34 and upload the following sketch to the Arduino 

connected to it. The sketch opens the BTserial at 9600 for communication mode. 

// Basic Bluetooth sketch HC-05_03_9600 

// Connect the HC-05 module and communicate using the serial monitor 

// 

// The HC-05 defaults to commincation mode when first powered on. 

// Needs to be placed in to AT mode 

// After a factory reset the default baud rate for communication mode is 38400 

// 

// 

//  Pins 

//  BT VCC to Arduino 5V out.  

//  BT GND to GND 

//  BT RX to Arduino pin 3 (through a voltage divider) 

//  BT TX to Arduino pin 2 (no need voltage divider) 

  

  

#include <SoftwareSerial.h> 

SoftwareSerial BTserial(2, 3); // RX | TX 

// Connect the HC-05 TX to Arduino pin 2 RX.  

// Connect the HC-05 RX to Arduino pin 3 TX through a voltage divider. 

  

char c = ' '; 

  

void setup()  

{ 

    // start the serial communication with the host computer 

    Serial.begin(9600); 

    Serial.println("Arduino with HC-05 is ready"); 

  

    // start communication with the HC-05 using 9600 

    BTserial.begin(9600);   

    Serial.println("BTserial started at 9600"); 

} 

  

void loop() 

{ 

  

     // Keep reading from HC-05 and send to Arduino Serial Monitor 

    if (BTserial.available()) 

    {   

        c = BTserial.read(); 

        Serial.write(c); 

    } 

    // Keep reading from Arduino Serial Monitor and send to HC-05 

    if (Serial.available()) 

    { 

        c =  Serial.read(); 

        BTserial.write(c);   

    } 

  

} 

Reset both Arduinos and remove power from the HC-05 and the HC-06 

Power on the HC-06. The LED should be rapidly blinking 5 times per second. This indicates it is 

waiting for a connection or to be paired. 

Power on the HC-05. The LED will blink a couple of times when first on and then change to a the 

regular pattern of blinking quickly every couple of seconds. 

The LED on the HC-06 will turn on full time (no blinking). 

The modules are now connected. 

Open the serial monitors and whatever you enter in one will appear the other and vice versa 

Every time the modules are powered on they should connect. If you find they do not connect then 

cycle the power to the Arduinos. Sometimes the HC-05 will get stuck trying to make a connection. In 

this case the LED on the HC-05 will either 

1. quickly blink 3 times every couple of seconds. 

2. blink on/off twice a second. 

This normally happens when the HC-06 is turned off and then back on again without resetting the 

HC-05. 

The devices seem to connect more quickly when the HC-06 is started first. 

Additional information 

For more information about AT commands look at the HC-05 user guide or the EGBT-046S/EGBT-

045MS user guide 

Phillipe Cantin has a similar article here  

Next step. Get Arduinos talking over Bluetooth. Arduino to Arduino by Bluetooth 

Arduino to Arduino by Bluetooth 

Posted on April 9, 2016  

Updated 12.06.2016: Added example 2 

In the Connecting 2 Arduinos by Bluetooth using a HC-05 and a HC-06: Pair, Bind, and Link post I 

explained how to connect a HC-05 to a HC-06 so that when powered they automatically made a 

connection. Here we look at using that connection to get Arduinos talking over Bluetooth. Before 

continuing you need to have the Arduinos and BT modules set up as per the previous post. Here I am 

using 2 HC-05s. One in master mode the other in slave mode. The setup process for the slave mode 

HC-05 is the same as the HC-06 in the previous post.  

http://elecfreaks.com/store/download/datasheet/Bluetooth/HC-0305%20serail%20module%20AT%20commamd%20set%20201104%20revised.pdf
https://docs.google.com/file/d/0BxdLxDCD6HidSkRaRTVuNERrQjg/edit
https://docs.google.com/file/d/0BxdLxDCD6HidSkRaRTVuNERrQjg/edit
http://phillipecantin.blogspot.hk/2014/08/hc-05-bluetooth-link-with-zero-code.html
http://www.martyncurrey.com/arduino-to-arduino-by-bluetooth/
http://www.martyncurrey.com/arduino-to-arduino-by-bluetooth/
http://www.martyncurrey.com/connecting-2-arduinos-by-bluetooth-using-a-hc-05-and-a-hc-06-pair-bind-and-link/

Set Up 

I am using 5V Arduino Nanos but any kind of 5V AVR based Arduino can be used. 

I have designated one of the Arduinos as the master device. This is the one that initiates the 

connection and in the first example it is the one that sends the commands. Having a master and 

slave setup makes the programming a little easier. 

To communicate with the BT modules I am using AltSoftSerial which uses pin 8 and pin 9. The 

AltSoftSerial library can be downloaded from 

https://www.pjrc.com/teensy/td_libs_AltSoftSerial.html and it will need to be added before you can 

compile the example sketches. 

Both BT modules are set with a communication baud rate of 9600. This can be changed but make 

sure you match the baud rate used when opening the software serial connection. 

    //  open software serial connection to the Bluetooth module. 

    BTserial.begin(9600); 

Connecting the Bluetooth Modules 

Most HC-05s and HC-06s have 3.3v TX and RX pins. 5V Arduinos will read 3.3v as HIGH so the BT 

modules TX pin can be connected directly to the Arduino RX pin. However, the Arduino TX pin needs 

to be converted to 3.3v before connecting to the BT modules RX pin. A simple way to do this is by 

using a voltage divider made from 2 resistors; I generally use 1 x 1K and 1 x 2K. 

Arduino RX (pin 8) to BT module TX pin 

Arduino TX (pin 9) to BT module RX pin via a voltage divider 

Both Arduinos have the same connections to the BT modules. 

https://www.pjrc.com/teensy/td_libs_AltSoftSerial.html
http://www.martyncurrey.com/wp-content/uploads/2016/04/Arduino2ArdionoBT_Breadboards_01_16001.jpg

Example 1: Remote Control an LED 

In the first example, we get one Arduino to control an LED connected to a second Arduino. 

Communication is one way only and there is no error checking. Arduino #1 simply sends the 

commands LEDON and LEDOFF to the Arduino #2. When the Arduino #2 gets the commands, it sets 

the LED accordingly. 

Example 1: Circuit 

Arduino #1, the master device, just has the Bluetooth module. Arduino #2, the slave device we have 

the Bluetooth module and an LED (with a suitable resistor) on pin D3. 

 

 

http://www.martyncurrey.com/wp-content/uploads/2016/04/Arduino2Arduino_1200.png
http://www.martyncurrey.com/wp-content/uploads/2016/04/Arduino2ArduinoExample1_circuit_1800.png
http://www.martyncurrey.com/wp-content/uploads/2016/04/Arduino2ArdionoBT_Breadboards_01_1600.jpg

Example 1: Sketches 

The Sketch on Arduino #1, the master device connected to the HC-05, simply sends the command 

LEDON, waits a second, sends the commands LEDOFF waits another second and then repeats 

indefinitely. 

/* 

* Sketch: Arduino2Arduino_MASTER_01 

* By Martyn Currey 

* 08.04.2016 

* Written in Arduino IDE 1.6.3 

* 

* Send commands through a serial connection to turn a LED on and OFF on a remote Arduino 

* There is no error checking and this sketch sends only 

* Commands should be contained within the start and end markers < and > 

* 

* D8 - AltSoftSerial RX 

* D9 - AltSoftSerial TX 

* 

*/ 

  

// AltSoftSerial uses D9 for TX and D8 for RX. While using AltSoftSerial D10 cannot be used for PWM. 

// Remember to use a voltage divider on the Arduino TX pin / Bluetooth RX pin 

// Download AltSoftSerial from https://www.pjrc.com/teensy/td_libs_AltSoftSerial.html 

  

#include <AltSoftSerial.h> 

AltSoftSerial BTserial;  

 

// Change DEBUG to true to output debug information to the serial monitor 

boolean DEBUG = true; 

  

void setup()   

{ 

    if (DEBUG) 

    { 

        // open serial communication for debugging and show  

        // the sketch filename and the date compiled 

        Serial.begin(9600); 

        Serial.println(__FILE__); 

        Serial.println(__DATE__); 

        Serial.println(" "); 

    } 

  

    //  open software serial connection to the Bluetooth module. 

    BTserial.begin(9600);  

    if (DEBUG)  {   Serial.println("BTserial started at 9600");     } 

  

} // void setup() 

  

  

void loop()   

{ 

    BTserial.println("<LEDON>"); 

    if (DEBUG) {Serial.println("LEDON command sent");}     

    delay (1000); 

  

    BTserial.println("<LEDOFF>"); 

    if (DEBUG) {Serial.println("LEDOFF command sent");}       

    delay (1000);     

} 

You will notice that the commands are contained within start and end markers, < and >. 

    BTserial.println("<LEDON>"); 

    BTserial.println("<LEDOFF>"); 

Using start and end markers allows the receiving device to check that it has a full command before 

acting upon it. 

The Sketch on Arduino #2, the slave device, checks for data and if there is a start marker it starts to 

put the recieved data in to the variable receivedChars[ ]. When an end marker is received it sets the 

newData flag to TRUE. Any data not within the start and end markers is ignored. 

When newData is TRUE we know we have a command to process. In this case we set pin D3 HIGH or 

LOW to turn the LED on or off. 

    if      (strcmp ("LEDON",receivedChars) == 0)  { digitalWrite(3,HIGH);   } 

    else if (strcmp ("LEDOFF",receivedChars) == 0) { digitalWrite(3,LOW);    } 

 

/* 

* Sketch: Arduino2Arduino_SLAVE_01 

* By Martyn Currey 

* 08.04.2016 

* Written in Arduino IDE 1.6.3 

* 

* Receive commands through a serial connection and turn a LED on or OFF 

* There is no error checking and this sketch receives only 

* Commands should be contained within the start and end markers < and > 

* 

* D8 - software serial RX 

* D9 - software serial TX 

* D3 - LED 

* 

*/ 

  

// AltSoftSerial uses D9 for TX and D8 for RX. While using AltSoftSerial D10 cannot be used for PWM. 

// Remember to use a voltage divider on the Arduino TX pin / Bluetooth RX pin 

// Download AltSoftSerial from https://www.pjrc.com/teensy/td_libs_AltSoftSerial.html 

#include <AltSoftSerial.h> 

AltSoftSerial BTserial;  

// Change DEBUG to true to output debug information to the serial monitor 

boolean DEBUG = true; 

// Variables used for incoming data 

const byte maxDataLength = 20;          // maxDataLength is the maximum length allowed for received data. 

char receivedChars[maxDataLength+1] ; 

boolean newData = false;               // newData is used to determine if there is a new command 

void setup()   

{ 

    // LED on pin 3 

    pinMode(3, OUTPUT);  

    digitalWrite(3,LOW); 

    if (DEBUG) 

    { 

        // open serial communication for debugging and show the sketch name and the date compiled 

        Serial.begin(9600); 

        Serial.println(__FILE__); 

        Serial.println(__DATE__); 

        Serial.println(" "); 

    } 

    //  open software serial connection to the Bluetooth module. 

    BTserial.begin(9600);  

    if (DEBUG)  {   Serial.println(F("AltSoftSerial started at 9600"));     } 

    newData = false; 

} // void setup() 

void loop()   

{ 

         recvWithStartEndMarkers();                // check to see if we have received any new commands 

         if (newData)  {   processCommand();  }    // if we have a new command do something 

} 

void processCommand() 

{ 

    newData = false; 

    if (DEBUG)  {   Serial.print("recieved data = ");  Serial.println(receivedChars);         } 

    if      (strcmp ("LEDON",receivedChars) == 0)  { digitalWrite(3,HIGH);   } 

    else if (strcmp ("LEDOFF",receivedChars) == 0) { digitalWrite(3,LOW);    } 

} 

// function recvWithStartEndMarkers by Robin2 of the Arduino forums 

// See  http://forum.arduino.cc/index.php?topic=288234.0 

void recvWithStartEndMarkers()  

{ 

     static boolean recvInProgress = false; 

     static byte ndx = 0; 

     char startMarker = '<'; 

     char endMarker = '>'; 

     if (BTserial.available() > 0)  

     { 

          char rc = BTserial.read(); 

          if (recvInProgress == true)  

          { 

               if (rc != endMarker)  

               { 

                    if (ndx < maxDataLength) { receivedChars[ndx] = rc; ndx++;  } 

               } 

               else  

               { 

                     receivedChars[ndx] = '\0'; // terminate the string 

                     recvInProgress = false; 

                     ndx = 0; 

                     newData = true; 

               } 

          } 

          else if (rc == startMarker) { recvInProgress = true; } 

     } 

} 

 

Example 2: Remote Temperature Monitor 

This example is a little more complex and unlike example 1, where the master device starts to send 

data as soon as it runs, this time it requests data from the slave. The slave receives the request from 

the master device and replies.  

 

The slave device waits until it it gets the “sendTemp” request, it then reads the value of the 

temperature sensor on pin A0, converts the value to degrees celsius and then sends it out as ascii via 

the software serial channel. The sketches use the same recvWithStartEndMarkers() function as 

before and so the data is wrapped in the < start marker and the > end marker. 

As before, the Bluetooth modules are set to automatically connect on start up. However, this time I 

am using 2 HC-05s; one in master mode, 1 in slave mode. 

On start-up, the master device initialises the LCD and starts a timer. When the timer finishes, it 

sends out a request for the temperature. The Timer is then reset. Within the main loop, the sketch is 

continuously checking for received data. When it receives a temperature, it displays it on the LCD. 

The slave simply sits and waits for the request. When it receives the request, it sends out the 

temperature and then goes back to waiting.  

Example 2: Circuit 

 

Example 2: Sketches 

The sketch on the master device is fairly simple. Once a second it sends a request to the slave device 

and waits for a reply. You can change the frequency of the request by changing the value of 

http://www.martyncurrey.com/wp-content/uploads/2016/04/Arduino2Ardiono_Example2_01_1600.jpg
http://www.martyncurrey.com/wp-content/uploads/2016/04/Arduino2ArduinoExample2Circuit_1800.png

waitTime. There is no error checking or time out and the sketch does not know if it ever gets a reply 

or not. 

/* 

* Sketch: Arduino2Arduino_example2_RemoteTemp_Master 

* By Martyn Currey 

* 11.05.2016 

* Written in Arduino IDE 1.6.3 

* 

* Send a temperature reading by Bluetooth 

* Uses the following pins 

* 

* D8 - software serial RX 

* D9 - software serial TX 

* A0 - single wire temperature sensor 

* 

* 

* AltSoftSerial uses D9 for TX and D8 for RX. While using AltSoftSerial D10 cannot be used for PWM. 

* Remember to use a voltage divider on the Arduino TX pin / Bluetooth RX pin 

* Download from https://www.pjrc.com/teensy/td_libs_AltSoftSerial.html 

*/ 

#include <AltSoftSerial.h> 

AltSoftSerial BTserial;  

// If you don't have an LCD you can use the serial monitor. 

#include <Wire.h>   

#include <LiquidCrystal_I2C.h> 

LiquidCrystal_I2C lcd(0x27, 2, 1, 0, 4, 5, 6, 7, 3, POSITIVE);  // Set the LCD I2C address 

// define the pattern for a custom degree character. You can also use chr 233 

byte degreeChr[8] = { B00111, B00101, B00111, B00000, B00000, B00000, B00000,};  

// Set DEBUG to true to output debug information to the serial monitor 

boolean DEBUG = true; 

// Variables used for incoming data 

const byte maxDataLength = 20; 

char receivedChars[21] ; 

boolean newData = false; 

// Variables used for the timer 

unsigned long startTime = 0; 

unsigned long waitTime = 1000; 

const byte TEMP_PIN = A0; 

void setup()   

{ 

  lcd.begin(20,4);  

  lcd.createChar(1, degreeChr); 

  lcd.setCursor(0,0);  lcd.print("Remote Temp Monitor");  

  lcd.setCursor(0,1);  lcd.print("Starting...");  

  if (DEBUG) 

  { 

       // open serial communication for debugging 

       Serial.begin(9600); 

       Serial.println(__FILE__); 

       Serial.println(" "); 

  } 

    BTserial.begin(9600);  

    if (DEBUG)  {  Serial.println("AltSoftSerial started at 9600");     } 

    newData = false;  

    startTime = millis(); 

} // void setup() 

void loop()   

{ 

    if (  millis()-startTime > waitTime )  

    { 

       BTserial.print("<sendTemp>");   

       if (DEBUG) { Serial.println("Request sent"); } 

       startTime = millis(); 

    } 

    recvWithStartEndMarkers();  

    if (newData)   

    {     

         if (DEBUG) { Serial.println("Data received"); } 

         lcd.setCursor(0,1);   lcd.print("Temp = ");  

         lcd.setCursor(7,1);   lcd.print(receivedChars);   

         lcd.write(1);  lcd.print("C");  

         newData = false;   

         receivedChars[0]='\0';      

    }     

} 

// function recvWithStartEndMarkers by Robin2 of the Arduino forums 

// See  http://forum.arduino.cc/index.php?topic=288234.0 

/* 

**************************************** 

* Function recvWithStartEndMarkers 

* reads serial data and returns the content between a start marker and an end marker. 

*  

* passed: 

*   

* global:  

*       receivedChars[] 

*       newData 

* 

* Returns: 

*           

* Sets: 

*       newData 

*       receivedChars 

* 

*/ 

void recvWithStartEndMarkers() 

{ 

     static boolean recvInProgress = false; 

     static byte ndx = 0; 

     char startMarker = '<'; 

     char endMarker = '>'; 

     char rc; 

     if (BTserial.available() > 0)  

     { 

          rc = BTserial.read(); 

          if (recvInProgress == true)  

          { 

               if (rc != endMarker)  

               { 

                    receivedChars[ndx] = rc; 

                    ndx++; 

                    if (ndx > maxDataLength) { ndx = maxDataLength; } 

               } 

               else  

               { 

                     receivedChars[ndx] = '\0'; // terminate the string 

                     recvInProgress = false; 

                     ndx = 0; 

                     newData = true; 

               } 

          } 

          else if (rc == startMarker) { recvInProgress = true; } 

     } 

} 

The slave device sketch is equally simple. It waits for a request, when it gets the request it sends out 

the current temperature. 

/* 

* Sketch: Arduino2Arduino_example2_RemoteTemp_Slave 

* By Martyn Currey 

* 11.05.2016 

* Written in Arduino IDE 1.6.3 

* 

* Send a temperature reading by Bluetooth 

* Uses the following pins 

* 

* D8 - software serial RX 

* D9 - software serial TX 

* A0 - single wire temperature sensor 

* 

* 

* AltSoftSerial uses D9 for TX and D8 for RX. While using AltSoftSerial D10 cannot be used for PWM. 

* Remember to use a voltage divider on the Arduino TX pin / Bluetooth RX pin 

* Download from https://www.pjrc.com/teensy/td_libs_AltSoftSerial.html 

*/ 

#include <AltSoftSerial.h> 

AltSoftSerial BTserial;  

// Set DEBUG to true to output debug information to the serial monitor 

boolean DEBUG = true; 

// Variables used for incoming data 

const byte maxDataLength = 20; 

char receivedChars[21] ; 

boolean newData = false; 

const byte TEMP_PIN = A0; 

void setup()   

{ 

    if (DEBUG) 

    { 

        // open serial communication for debugging 

        Serial.begin(9600); 

        Serial.println(__FILE__); 

        Serial.println(" "); 

    } 

    //  open a software serial connection to the Bluetooth module. 

    BTserial.begin(9600);  

    if (DEBUG)  {   Serial.println(F("AltSoftSerial started at 9600"));     } 

    newData = false; 

} // void setup() 

void loop()   

{ 

    recvWithStartEndMarkers();  

    if (newData)  {   processCommand();  }     

} 

/* 

**************************************** 

* Function getTemp 

* read a analogue pin and converts the value to a temperature 

* based on the adafruit thermistor guide https://learn.adafruit.com/thermistor/testing-a-thermistor  

*  

* passed: 

*   

* global:  

* 

* Returns: 

*        float temp   

* Sets: 

* 

*/ 

float getTemp() 

{ 

    float reading = analogRead(TEMP_PIN); 

    reading = 1023 / reading - 1; 

    reading = 10000  / reading; 

    float steinhart; 

    steinhart = reading / 10000;          // (R/Ro) 

    steinhart = log(steinhart);           // ln(R/Ro) 

    steinhart /= 3950;                    // 1/B * ln(R/Ro) 

    steinhart += 1.0 / (25 + 273.15);     // + (1/To) 

    steinhart = 1.0 / steinhart;          // Invert 

    steinhart -= 273.15;                  // convert to C 

    return steinhart; 

  

} 

/* 

**************************************** 

* Function processCommand 

* parses data commands contained in receivedChars[] 

* receivedChars[] has not been checked for errors 

*  

* passed: 

*   

* global:  

*       receivedChars[] 

*       newData 

* 

* Returns: 

*           

* Sets: 

*       receivedChars[] 

*       newData 

* 

*/ 

void processCommand() 

{ 

  

     Serial.println(receivedChars); 

  

     if (strcmp ("sendTemp",receivedChars) == 0)  

     {  

         float temp = getTemp(); 

         BTserial.print("<");  BTserial.print( temp ); BTserial.print(">"); 

         if (DEBUG) { Serial.print("Temp is ");   Serial.print(temp);  Serial.println("");   } 

     } 

  

     newData = false; 

     receivedChars[0]='\0';  

  

} 

// function recvWithStartEndMarkers by Robin2 of the Arduino forums 

// See  http://forum.arduino.cc/index.php?topic=288234.0 

/* 

**************************************** 

* Function recvWithStartEndMarkers 

* reads serial data and returns the content between a start marker and an end marker. 

*  

* passed: 

*   

* global:  

*       receivedChars[] 

*       newData 

* 

* Returns: 

*           

* Sets: 

*       newData 

*       receivedChars 

* 

*/ 

void recvWithStartEndMarkers() 

{ 

     static boolean recvInProgress = false; 

     static byte ndx = 0; 

     char startMarker = '<'; 

     char endMarker = '>'; 

     char rc; 

     if (BTserial.available() > 0)  

     { 

          rc = BTserial.read(); 

          if (recvInProgress == true)  

          { 

               if (rc != endMarker)  

               { 

                    receivedChars[ndx] = rc; 

                    ndx++; 

                    if (ndx > maxDataLength) { ndx = maxDataLength; } 

               } 

               else  

               { 

                     receivedChars[ndx] = '\0'; // terminate the string 

                     recvInProgress = false; 

                     ndx = 0; 

                     newData = true; 

               } 

          } 

          else if (rc == startMarker) { recvInProgress = true; } 

     } 

} 

  

 



http://www.martyncurrey.com/connecting-2-arduinos-by-bluetooth-using-a-hc-05-and-a-hc-06-
pair-bind-and-link/ 

Page 24 of 24 

This entry was posted in Arduino, Bluetooth and tagged HC-05, HC-06 by Martyn. Bookmark the 

permalink. 

http://www.martyncurrey.com/category/arduino/
http://www.martyncurrey.com/category/bluetooth/
http://www.martyncurrey.com/tag/hc-05/
http://www.martyncurrey.com/tag/hc-06/
http://www.martyncurrey.com/author/admin/
http://www.martyncurrey.com/arduino-to-arduino-by-bluetooth/

